Université Grenoble Alpes Année 2016-2017
Module MAT 307

CC1 : examen partiel du 2 novembre 2016

Une feuille A4 recto-verso <u>*manuscrite*</u> *est autorisée*

Calculatrices autorisées

Le barème est donné à titre indicatif.

Durée 2h

Exercice – [*? ? points*]

On considère la courbe paramétrée en polaire $r(\theta) = -\ln(\cos(\theta))$ pour $\theta \in \mathbb{R}$.

1. **Donner le domaine de définition et réduire l'intervalle d'étude par un argument de périodicité.**
   r est périodique de période 2π, et on peut se ramener à une étude sur $[0, \pi]$ par parité. Sur cet intervalle, $r(\theta)$ est définie si $\cos(\theta) \geq 0$ donc pour $\theta \in [0, \pi/2]$.
2. **Etudier les variations de $r(\theta)$ pour $\theta \in]0, \pi/2[$.**
   cos est décroissante et $-\ln$ décroissante, donc r est croissante de $r(0) = 0$ à $\lim_{\theta \to \pi/2} r = +\infty$. On peut aussi calculer $r' = \sin(\theta)/\cos(\theta)$.
   Remarque : on observe que $r(0) = r'(0) = 0$, donc l'origine est un point singulier, comme r reste positif il s'agit d'un point de rebroussement, avec tangente horizontale (car faisant un angle $\theta = 0$ avec l'axe des x).
3. **Trouver la valeur de $\theta \in]0, \pi/2[$ pour laquelle la distance du point $(r(\theta)\cos(\theta), r(\theta)\sin(\theta))$ à l'axe (Oy) est maximale. Donner une valeur approchée.**
   (Indication par rapport au tracé : quelle est la direction de la tangente en ce point ?)
   $x = -\ln(\cos(\theta))\cos(theta)$ est maximal lorsque $x' = 0$. Or

   $$x' = \frac{\sin(\theta)}{\cos(\theta)}\cos(\theta) + \ln(\cos(\theta)\sin(\theta) = \sin(\theta)(1 + \ln(\cos(\theta)))$$

   Donc x' s'annule sur $]0, \pi/2[$ en θ tel que $1 + \ln(\cos(\theta)) = 0$ soit $\cos(\theta) = e^{-1}$ ou $\theta = \mathrm{acos}(1/e)$. En ce point $r = 1$ (et $x = 1/e$).

   ```
   X:=-ln(cos(x))*cos(x); dX:=factor(X');
   ```

   $$-\ln(\cos(x))\cos(x), (\ln(\cos(x)) + 1)\sin(x)$$

   ```
   [t]:=solve(dX=0,x)|(x>0 and x<pi/2); Xm:=X(x=t);
   ```

   $$\{\mathrm{acos}\left(\mathrm{inv}\left(e^1\right)\right)\}, \mathrm{inv}\left(e^1\right)$$

   Les valeurs approchées :

   ```
   evalf(t,4); evalf(Xm,4);
   ```

   $$1.194, 0.3679$$

4. **Etudier la branche infinie.**
   En $\theta = \pi/2$, le logarithme tend vers l'infini, il y a une branche infinie. On recherche

l'asymptote dans le repère tourné de $\theta = \pi/2$ en multipliant $r(\theta)$ par $\sin(\theta - \pi/2)$ avec θ tendant vers $\pi/2^-$ On pose $\theta = \pi/2 - h$ avec h tendant vers 0^+.

$$-\ln(\cos(\frac{\pi}{2} - h)) * \sin(-h) = \sin(h)\ln(\sin(h))$$

Comme $\sin(h)$ est équivalent à h en 0 et comme $\ln(h)$ tend vers l'infini beaucoup plus lentement que h ne tend vers 0 ($\lim h\ln(h) = 0$ en 0) la limite est nulle. On peut vérifier à la machine :

```
limit(-ln(cos(x))*sin(x-pi/2),x=pi/2)
```

$$0$$

On a donc une asymptote d'équation $Y = 0$ dans le repère tourné de $\pi/2$, c'est l'axe Oy dans le repère non tourné.

5. **Tracer la courbe pour $\theta \in [0, \pi/2[$ (faire apparaitre sur la courbe le point de paramètre $\theta = 0$ et le sens de parcours).**
   ```
   plotpolar(-ln(cos(x)),x=0..pi/2)
   ```



6. **En utilisant un argument de symétrie et le tracé de la question précédente, tracer la courbe $r(\theta) = -\ln(\cos(\theta))$ pour $\theta \in]-\pi/2, \pi/2[$.**
   On a une symétrie par rapport à Ox puisque $r(-\theta) = r(\theta)$.
   ```
   plotpolar(-ln(cos(x)),x=-pi/2..pi/2)
   ```



Problème – [*? ? points*]
On considère la courbe paramétrée $(x(t), y(t)) = (|t^2 - 1|^{3/2}, t^3)$, $t \in \mathbb{R}$.
(Rappel : $(1+h)^\alpha = 1 + \alpha h + \mathcal{O}(h^2)$ *avec* $\alpha \in \mathbb{R}$ *fixé* et $h \to 0$.)

1. **Donner le domaine de définition, et réduire l'intervalle d'étude par un argument de symétrie.**
   x et y sont définis sur $\mathbb{R}$. x est paire, y est impaire, donc symétrie par rapport à Ox. On se ramène à $[0, +\infty[$.
2. **Etude pour $t \in [0, 1]$:**
   (a) **En écrivant $x(t)$ sans les valeurs absolues, calculer $x'(t)$.**
   On a $x(t) = (1 - t^2)^{3/2}$, donc
   $$x'(t) = \frac{3}{2}(-2t)(1-t^2)^{1/2} = -3t\sqrt{1-t^2}, \quad y' = 3t^2$$
   (b) **Calculer $\lim\limits_{t \to 1-} x'(t)$ et en déduire la direction de la tangente en $t = 1$.**
   $x' \to 0$, $y' = 3$ donc tangente verticale.
   (c) **Donner le double tableau de variations sur $[0, 1]$.**
   x décroit de $x(0) = 1$ à $x(1) = 0$, y croit de 0 à 1
   (d) **Y a t-il des points singuliers ? BONUS : Si oui, étudier les (nature et tangente).**
   x' et y' s'annulent simultanément en $t = 0$ seul point d'annulation de y', donc il y a un point singulier $(1, 0)$ de paramètre $t = 0$. En utilisant le DL rappelé dans l'énoncé on a $x = 1 - \frac{3}{2}t^2 + O(t^4)$ et $y = t^3$ donc la tangente est horizontale et c'est un rebroussement (de 1ère espèce). On peut aussi calculer
   $$x'' = -3\sqrt{1-t^2} + \frac{3t^2}{\sqrt{1-t^2}} = \frac{6t^2 - 3}{\sqrt{1-t^2}}, \quad y'' = 6t$$
   $x''(0) = -3$ et $y''(0) = 0$. On peut vérifier par un argument de symétrie en 0, la tangente est symétrique par rapport à Ox.
   (e) **Tracer la courbe (faire apparaitre sur la courbe les points de paramètres t=0, 1 et le sens de parcours).**

```
purge(t);G:=plotparam([(1-t^2)^(3/2),t^3],t,0,1);
```



(f) **Calculer la longueur d'arc entre les points de paramètres $t = 0$ et $t = 1$.**

$$\sqrt{x'^2 + y'^2} = \sqrt{(-3t\sqrt{1-t^2})^2 + (3t^2)^2} = \sqrt{9t^2(1-t^2+t^2)} = 3t$$

Donc la longueur demandée vaut

$$\int_0^1 \sqrt{x'^2 + y'^2} dt = \int_0^1 3t \; dt = \frac{3}{2}$$

Vérification sur la courbe cela parait plausible

(g) **En $t = 1/2$, donner le repère de Frénet, la courbure et le centre du cercle osculateur. Rajouter le tracé du cercle osculateur dans le tracé précédent.**
Le vecteur tangent est

$$\overrightarrow{T} = \frac{(x', y')}{\sqrt{x'^2 + y'^2}} = (-\sqrt{1-t^2}, t)$$

on retrouve bien un vecteur horizontal en $t = 0$ (dirigé vers la gauche). En $t = 1/2$, $\overrightarrow{T} = (-\sqrt{3}/2, 1/2)$, qui dirige la tangente au point de coordonnées :

$$((1-t^2)^{3/2}, t^3) = (\frac{3\sqrt{3}}{8}, \frac{1}{8})$$

Le vecteur normal est $\overrightarrow{N} = (-t, -\sqrt{1-t^2})$, qui donne en $t = 1/2$ $\overrightarrow{N} = (-1/2, -\sqrt{3}/2)$.
On a

$$\kappa \overrightarrow{N} = \frac{dT}{ds} = \frac{dt}{ds}\frac{dT}{dt}$$

soit

$$\kappa(-t, -\sqrt{1-t^2}) = \frac{1}{3t}(\frac{t}{\sqrt{1-t^2}}, 1)$$

donc

$$\kappa = -\frac{1}{3t\sqrt{1-t^2}} = -\frac{4\sqrt{3}}{9}$$

Autre méthode, la courbure est aussi donnée par

$$\kappa = \frac{x'y'' - x''y'}{(3t)^3} = -\frac{1}{3t\sqrt{1-t^2}}$$

Vérification :

```
assume(t>0);k:=normal(curvature(G,t));normal
(k(t=1/2));
```

$$t, \frac{\sqrt{-t^2+1}}{(3\cdot t^3 - 3\cdot t)}, -\frac{4\cdot\sqrt{3}}{9}$$

Le centre du cercle osculateur est donné par

$$((1-t^2)^{3/2}, t^3) + \frac{1}{\kappa}N$$

en $t = 1/2$ soit

$$(\frac{3\sqrt{3}}{8}, \frac{1}{8}) - \frac{9}{4\sqrt{3}}(-\frac{1}{2}, -\frac{\sqrt{3}}{2}) = (\frac{3\sqrt{3}}{4}, \frac{5}{4})$$

```
gl_x=-2..3;gl_ortho=true;XY:=[(1-t^2)^(3/2
),t^3];G:=plotparam(XY,t,0,1); M:=point(XY
(t=1/2));C:=osculating_circle(G,1/2,color=blue
); I:=center(C);
```



Vérification des coordonnées du centre du cercle :

```
normal(coordinates(I));
```

$$[\frac{3\cdot\sqrt{3}}{4}, \frac{5}{4}]$$

---

3. **Etude pour $t \in [1, +\infty[$:**

   (a) **En écrivant $x(t)$ sans les valeurs absolues, calculer $x'(t)$.**
   $x(t) = (t^2-1)^{3/2}$ donc $x'(t) = 3t\sqrt{t^2-1}$

(b) **Montrer que** $\lim_{t\to 1+} x'(t) = \lim_{t\to 1-} x'(t)$**.**
0=0

(c) **Donner le double tableau de variations sur** $[1, +\infty[$**.**
x croit de 0 à $+\infty$ et y de 1 à $+\infty$

(d) **Y a t-il des points singuliers ? BONUS : Si oui, étudier les (nature et tangente).**
Non, puisque $y' = 3t^2$ ne s'annule pas si $t \geq 1$. Zut, pas de bonus :-(

(e) **Etudier les branches infinies.** *(Indication : observer que pour* $t \to +\infty$, $x(t) = t^3(1-\frac{1}{t^2})^{3/2}$ *avec* $h = -1/t^2 \to 0$.*)*
Lorsque $t \to +\infty$, x et y tendent vers l'infini, on calcule y/x avec l'indication

$$\frac{y}{x} = (1-\frac{1}{t^2})^{-3/2} \to 1$$

puis

$$y - x = t^3(1-(1+h)^{3/2}) = t^3(-\frac{3}{2}h + O(h^2)) = \frac{3}{2}t + O(\frac{1}{t}) \to +\infty$$

Branche parabolique de direction asymptotique $y = x$.

4. **Tracer la courbe pour tout t dans le domaine de définition (faire apparaitre sur la courbe les points de paramètres t=0, 1 et le sens de parcours).**

```
plotparam([abs(t^2-1)^(3/2),t^3],t,-2,2)
```



5. **Calculer la courbure pour $t > 1$.**
(Consigne : on pourra donner le résultat du calcul fait à la calculatrice à condition de donner la commande utilisée)

```
assume(t>1);X:=(t^2-1)^(3/2); Y:=t^3;
```

$$t, \sqrt{t^2-1}(t^2-1), t^3$$

```
X1:=simplify(diff(X,t)); Y1:=simplify(diff(Y,t))
```

$$3 \cdot t\sqrt{t^2-1}, 3 \cdot t^2$$

```
X2:=simplify(diff(X1,t)); Y2:=simplify(diff(Y1,t))
```

$$\frac{(6 \cdot t^2\sqrt{t^2-1} - 3\sqrt{t^2-1})}{(t^2-1)}, 6 \cdot t$$

```
k:=simplify((X1*Y2-X2*Y1)/sqrt(X1^2+Y1^2)^3);
```

$$-\frac{\sqrt{2 \cdot t^4 - 3 \cdot t^2 + 1}}{(12 \cdot t^7 - 24 \cdot t^5 + 15 \cdot t^3 - 3 \cdot t)}$$

Vérification

```
simplify(curvature(plotparam([(t^2-1)^(3/2
),t^3]),t))
```

$$-\frac{\sqrt{2 \cdot t^4 - 3 \cdot t^2 + 1}}{(12 \cdot t^7 - 24 \cdot t^5 + 15 \cdot t^3 - 3 \cdot t)}$$

Autre forme du résultat :

```
1/factor(simplify(1/k))
```

$$-\frac{1}{3 \cdot t(2 \cdot t^2 - 1)\sqrt{2 \cdot t^4 - 3 \cdot t^2 + 1}}$$

ou encore en factorisant partiellement sous la racine :

$$-\frac{1}{3t(2t^2-1)\sqrt{(t^2-1)(2t^2-1)}}$$